# $Id$
# Maintainer: Eric Bélanger <eric@archlinux.org>

pkgname=audacity
pkgver=2.1.3
pkgrel=1
pkgdesc="A program that lets you manipulate digital audio waveforms"
arch=('i686' 'x86_64')
url="http://audacityteam.org"
license=('GPL')
depends=('libmad' 'libid3tag' 'wxgtk' 'lame' 'lilv' 'soundtouch'
         'ffmpeg' 'vamp-plugin-sdk' 'sbsms' 'portsmf' 'desktop-file-utils')
makedepends=('cmake' 'python2')
options=('!makeflags')
source=(https://github.com/audacity/audacity/archive/Audacity-${pkgver}.zip)
sha1sums=('616097deb29d3883ca2d858fcefda1550cdfbdf2')

build() {
  cd audacity-Audacity-${pkgver}
  autoreconf -vi # use system libraries
  ./configure --prefix=/usr --with-libsamplerate
  make
}

package() {
  cd audacity-Audacity-${pkgver}
  make DESTDIR="${pkgdir}" install
}
